

EXCEL



By Bill Jelen

SMALL and LARGE Functions

A college professor in my Power Excel seminar at IMA's 90th Annual Conference & Exposition in Denver posed the question of how to identify the two lowest scores in a range so that they could be removed from the final grade. Finding the lowest score is easy with MIN, but what about the second lowest?

Most people are familiar with the MIN and MAX functions in Excel. Microsoft considers these functions, along with SUM, COUNT, and AVERAGE, to be important enough to offer in the Auto-Sum icon's dropdown box. While MIN can find the smallest value in a range, it can't help if you need the second-smallest value. For that, you'll want to learn about the SMALL function. This month's article will show you uses for SMALL and LARGE functions, from finding outliers to sorting with a formula.

Identify the Second-Smallest Value in a Range

You have a series of amounts in cells A2:A20. Using =MIN(A2:A20) will get you the smallest value. When it comes to the second-smallest value, however, the SMALL function steps in: The formula in

this instance is =SMALL(A2:A20,2). Similarly, the largest value can be found with =MAX(A2:A20), while =LARGE(A2:A20,3) will get you the third-largest value.

Note that you can also use SMALL and LARGE to replace MIN and MAX. Using =SMALL(A2:A20,1) will return the minimum value, and =LARGE(A2:A20,1) will return the maximum value. I don't recommend this as a general practice because MIN and MAX calculate 2.5 times faster than SMALL and LARGE. If you are trying to keep a range of formulas consistent, however, then it is appropriate. See cells F5:F9 of Figure 1 as an example.

The SMALL and LARGE functions will always return a specific value found in the range. In most cases, this is better than using the PERCENTILE or QUARTILE functions, as these functions return a statistical number that may not occur in the

Figure 1

	F5		fx =SMALL(\$A\$2:\$A\$20,E5)								
	A	B	C	D	E	F	G	H	I	J	K
			Percent Rank			Min			Max		Sorted
1	Amount	Rank									
2	1728	6	72.2%			154			2820		154
3	606	17	11.1%								445
4	1214	12	38.8%								606
5	1529	8	61.1%		1	154		1	2820		718
6	796	15	22.2%		2	445		2	2789		796
7	2309	4	83.3%			606		3	2705		985
8	985	13	27.7%			718		4	2309		985
9	2705	3	88.8%			796		5	2241		1214
10	1704	7	66.6%								1365
11	1452	9	55.5%								1446
12	2789	2	94.4%		0.05	415.9		1	890.5		1452
13	985	13	27.7%		0.25	890.5		2	1446		1529
14	1365	11	44.4%		0.5	1446		3	1984.5		1704
15	445	18	5.5%		0.75	1984.5		4	2820		1728
16	718	16	16.6%		0.95	2792.1					2241
17	154	19	0.0%		0.99	2814.42					2309
18	2241	5	77.7%								2705
19	2820	1	100.0%								2789
20	1446	10	50.0%			1473.2105					2820

actual data set. For example, cell F9 calculates that the fifth-smallest number in the range is 796. You can rely that this number will always be located in the original range (in this case, in cell A6). Conversely, the 890.5 calculated by the PERCENTILE function in F13 and the QUARTILE function in cell I12 aren't numbers located in the original range. If you're planning on using VLOOKUP or MATCH to grab additional data from the original unsorted data set, it's critical that the value can be found in the original data.

Sorting with a Formula

The SMALL function also provides a simple solution to the problem of how to sort a range with a formula. Typically, people will try to solve this using a series of RANK formulas to assign ranks of 1 to 19 to each number in the original data set. You can then use VLOOKUP if the ranks are to the left of the data or INDEX and MATCH if the ranks are to the right of the data. But these methods fail when a tie is encountered. In Figure

1, two cells have a rank of 13, and no cells have a rank of 14. This is due to two different cells being tied with a value of 985. The missing rank of 14 will cause problems when you try to use a VLOOKUP to return the 14th value in the data set.

SMALL and LARGE don't have the same problem with ties. Both =SMALL(A2:A20,6) and =SMALL(A2:A20,7) will return the value of 985. This makes SMALL a great solution for returning a sorted list of numbers from the original data set.

Replacing a Range of 1, 2, 3 with ROW

You may often find that you create a small range of sequential numbers to be used as arguments in an adjacent formula. In Figure 1, the numbers in cells E5:E9 are used as the second argument in the formulas in cells F5:F9. Usually, the sorted list in column K would need an adjacent range with the numbers from 1 to 19. You

can eliminate the need for the range of numbers by using the ROW function. For example, if you ask for the ROW(A1), Excel will return the number 1. If you copy this formula down several rows, the formula will change to ROW(A2), ROW(A3) and so on. This effectively returns the numbers 1, then 2, then 3.

Usually, the formula in cell K2 would have been =SMALL(\$A\$2:\$A\$20,1), and the formula in cell K3 would have been =SMALL(\$A\$2:\$A\$20,2). It would be tedious to edit the second argument in each individual formula. Instead, enter =SMALL(\$A\$2:\$A\$20,ROW(A1)) in cell K2 and copy it down. Since ROW(A1) evaluates to 1, you'll get the smallest value in K2. The copied formula will ask for ROW(A2) in cell K3, which will give you the second-smallest value, and so on.

From sorting lists to finding outliers, SMALL and LARGE are great tools to have in your Excel arsenal. **SF**

Bill Jelen is the host of MrExcel.com and a frequent speaker at IMA chapter professional development days. Send questions for future articles to IMA@MrExcel.com.

Table 1. Formulas in Figure 1

B2: =RANK(A2,\$A\$2:\$A\$20)

C2: =PERCENTRANK(\$A\$2:\$A\$20,A2)

F2: =MIN(A2:A20)

I2: =MAX(A2:A20)

F5: =SMALL(\$A\$2:\$A\$20,E5)

I5: =LARGE(\$A\$2:\$A\$20,H5)

F12: =PERCENTILE(\$A\$2:\$A\$20,E12)

I12: =QUARTILE(\$A\$2:\$A\$20,H12)

F20: =AVERAGE(A2:A20)

K2: =SMALL(\$A\$2:\$A\$20,ROW(A1))